

# NAG Fortran Library Routine Document

## F08HBF (DSBEVX)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F08HBF (DSBEVX) computes selected eigenvalues and, optionally, eigenvectors of a real  $n$  by  $n$  symmetric band matrix  $A$  of bandwidth  $(2k_d + 1)$ . Eigenvalues and eigenvectors can be selected by specifying either a range of values or a range of indices for the desired eigenvalues.

### 2 Specification

```

SUBROUTINE F08HBF (JOBZ, RANGE, UPLO, N, KD, AB, LDAB, Q, LDQ, VL, VU,
1      IL, IU, ABSTOL, M, W, Z, LDZ, WORK, IWORK, JFAIL,
2      INFO)
      INTEGER      N, KD, LDAB, LDQ, IL, IU, M, LDZ, IWORK(*), JFAIL(*),
1      INFO
      double precision AB(LDAB,*), Q(LDQ,*), VL, VU, ABSTOL, W(*), Z(LDZ,*),
1      WORK(*)
      CHARACTER*1  JOBZ, RANGE, UPLO

```

The routine may be called by its LAPACK name ***dsbev***x.

### 3 Description

The symmetric band matrix  $A$  is first reduced to tridiagonal form, using orthogonal similarity transformations. The required eigenvalues and eigenvectors are then computed from the tridiagonal matrix; the method used depends upon whether all, or selected, eigenvalues and eigenvectors are required.

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Demmel J W and Kahan W (1990) Accurate singular values of bidiagonal matrices *SIAM J. Sci. Statist. Comput.* **11** 873–912

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

- 1: JOBZ – CHARACTER\*1 *Input*  
*On entry:* if JOBZ = 'N', compute eigenvalues only.  
 If JOBZ = 'V', compute eigenvalues and eigenvectors.  
*Constraint:* JOBZ = 'N' or 'V'.
- 2: RANGE – CHARACTER\*1 *Input*  
*On entry:* if RANGE = 'A', all eigenvalues will be found.  
 If RANGE = 'V', all eigenvalues in the half-open interval (VL, VU] will be found.  
 If RANGE = 'I', the ILth to IUth eigenvalues will be found.

- 3: UPLO – CHARACTER\*1 *Input*  
*On entry:* if UPLO = 'U', the upper triangle of  $A$  is stored.  
 If UPLO = 'L', the lower triangle of  $A$  is stored.
- 4: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 5: KD – INTEGER *Input*  
*On entry:*  $k_d$ , the number of super-diagonals of the matrix  $A$  if UPLO = 'U', or the number of sub-diagonals if UPLO = 'L'.  
*Constraint:*  $KD \geq 0$ .
- 6: AB(LDAB,\*) – **double precision** array *Input/Output*  
**Note:** the second dimension of the array AB must be at least  $\max(1, N)$ .  
*On entry:* the upper or lower triangle of the symmetric band matrix  $A$ , stored in the first  $KD + 1$  rows of the array. The  $j$ th column of  $A$  is stored in the  $j$ th column of the array AB as follows:  
     if UPLO = 'U',  $AB(k_d + 1 + i - j, j) = a_{ij}$  for  $\max(1, j - k_d) \leq i \leq j$ ;  
     if UPLO = 'L',  $AB(1 + i - j, j) = a_{ij}$  for  $j \leq i \leq \min(n, j + k_d)$ .  
*On exit:* is overwritten by values generated during the reduction to tridiagonal form. If UPLO = 'U', the first super-diagonal and the diagonal of the tridiagonal matrix  $T$  are returned in rows  $KD$  and  $KD + 1$  of AB, and if UPLO = 'L', the diagonal and first sub-diagonal of  $T$  are returned in the first two rows of AB.
- 7: LDAB – INTEGER *Input*  
*On entry:* the first dimension of the array AB as declared in the (sub)program from which F08HBF (DSBEVX) is called.  
*Constraint:*  $LDAB \geq KD + 1$ .
- 8: Q(LDQ,\*) – **double precision** array *Output*  
**Note:** the second dimension of the array Q must be at least  $\max(1, N)$ .  
*On exit:* if JOBZ = 'V', the  $n$  by  $n$  orthogonal matrix used in the reduction to tridiagonal form.  
 If JOBZ = 'N', the array Q is not referenced.
- 9: LDQ – INTEGER *Input*  
*On entry:* the first dimension of the array Q as declared in the (sub)program from which F08HBF (DSBEVX) is called.  
*Constraint:* if JOBZ = 'V',  $LDQ \geq \max(1, N)$ .
- 10: VL – **double precision** *Input*  
 11: VU – **double precision** *Input*  
*On entry:* if RANGE = 'V', the lower and upper bounds of the interval to be searched for eigenvalues.  
*Constraint:*  $VL < VU$ .  
 If RANGE = 'A' or 'T', VL and VU are not referenced.

- 12: IL – INTEGER *Input*  
 13: IU – INTEGER *Input*

*On entry:* if RANGE = 'I', the indices (in ascending order) of the smallest and largest eigenvalues to be returned.

If RANGE = 'A' or 'V', IL and IU are not referenced.

*Constraints:*

if  $N = 0$ ,  $IL = 1$  and  $IU = 0$ ;  
 if  $N > 0$ ,  $1 \leq IL \leq IU \leq N$ .

- 14: ABSTOL – **double precision** *Input*

*On entry:* the absolute error tolerance for the eigenvalues. An approximate eigenvalue is accepted as converged when it is determined to lie in an interval  $[a, b]$  of width less than or equal to

$$ABSTOL + \epsilon \max(|a|, |b|),$$

where  $\epsilon$  is the **machine precision**. If ABSTOL is less than or equal to zero, then  $\epsilon \|T\|_1$  will be used in its place, where  $T$  is the tridiagonal matrix obtained by reducing  $A$  to tridiagonal form. Eigenvalues will be computed most accurately when ABSTOL is set to twice the underflow threshold  $2 \times X02AMF()$ , not zero. If this routine returns with INFO > 0, indicating that some eigenvectors did not converge, try setting ABSTOL to  $2 \times X02AMF()$ . See Demmel and Kahan (1990).

- 15: M – INTEGER *Output*

*On exit:* the total number of eigenvalues found.

If RANGE = 'A',  $M = N$ .

If RANGE = 'I',  $M = IU - IL + 1$ .

*Constraint:*  $0 \leq M \leq N$ .

- 16: W(\*) – **double precision** array *Output*

**Note:** the dimension of the array W must be at least  $\max(1, N)$ .

*On exit:* the first M elements contain the selected eigenvalues in ascending order.

- 17: Z(LDZ,\*) – **double precision** array *Output*

**Note:** the second dimension of the array Z must be at least  $\max(1, M)$ .

*On exit:* if JOBZ = 'V', then if INFO = 0, the first  $m$  columns of  $Z$  contain the orthonormal eigenvectors of the matrix  $A$  corresponding to the selected eigenvalues, with the  $i$ th column of  $Z$  holding the eigenvector associated with  $W(i)$ .

If an eigenvector fails to converge, then that column of  $Z$  contains the latest approximation to the eigenvector, and the index of the eigenvector is returned in JFAIL.

If JOBZ = 'N',  $Z$  is not referenced.

**Note:** the user must ensure that at least  $\max(1, M)$  columns are supplied in the array  $Z$ ; if RANGE = 'V', the exact value of  $M$  is not known in advance and an upper bound must be used.

- 18: LDZ – INTEGER *Input*

*On entry:* the first dimension of the array  $Z$  as declared in the (sub)program from which F08HBF (DSBEVX) is called.

*Constraints:*

if JOBZ = 'V',  $LDZ \geq \max(1, N)$ ;  
 $LDZ \geq 1$  otherwise.

- 19: WORK(\*) – **double precision** array Workspace  
**Note:** the dimension of the array WORK must be at least  $\max(1, 7 \times N)$ .
- 20: IWORK(\*) – INTEGER array Workspace  
**Note:** the dimension of the array IWORK must be at least  $\max(1, 5 \times N)$ .
- 21: JFAIL(\*) – INTEGER array Output  
**Note:** the dimension of the array JFAIL must be at least  $\max(1, N)$ .  
*On exit:* if JOBZ = 'V', then if INFO = 0, the first M elements of JFAIL are zero. If INFO > 0, JFAIL contains the indices of the eigenvectors that failed to converge.  
 If JOBZ = 'N', JFAIL is not referenced.
- 22: INFO – INTEGER Output  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th argument had an illegal value.

INFO > 0

If INFO =  $i$ , then  $i$  eigenvectors failed to converge. Their indices are stored in array JFAIL. Please see ABSTOL.

## 7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix  $(A + E)$ , where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and  $\epsilon$  is the **machine precision**. See Section 4.7 of Anderson *et al.* (1999) for further details.

## 8 Further Comments

The total number of floating point operations is proportional to  $k_d n^2$  if JOBZ = 'N', and is proportional to  $n^3$  if JOBZ = 'V' and RANGE = 'A', otherwise the number of floating point operations will depend upon the number of computed eigenvectors.

The complex analogue of this routine is F08HPF (ZHBEVX).

## 9 Example

To find the eigenvalues in the half-open interval  $(-3, 3]$ , and the corresponding eigenvectors, of the symmetric band matrix

$$A = \begin{pmatrix} 1 & 2 & 3 & 0 & 0 \\ 2 & 2 & 3 & 4 & 0 \\ 3 & 3 & 3 & 4 & 5 \\ 0 & 4 & 4 & 4 & 5 \\ 0 & 0 & 5 & 5 & 5 \end{pmatrix}.$$

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F08HBF Example Program Text
*      Mark 21. NAG Copyright 2004.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          KDMAX, NMAX, MMAX
PARAMETER        (KDMAX=5,NMAX=20,MMAX=10)
INTEGER          LDAB, LDQ, LDZ
PARAMETER        (LDAB=KDMAX+1,LDQ=NMAX,LDZ=NMAX)
CHARACTER        UPLO
PARAMETER        (UPLO='U')
DOUBLE PRECISION ZERO
PARAMETER        (ZERO=0.0D+0)
*      .. Local Scalars ..
DOUBLE PRECISION ABSTOL, VL, VU
INTEGER          I, IFAIL, IL, INFO, IU, J, KD, M, N
*      .. Local Arrays ..
DOUBLE PRECISION AB(LDAB,NMAX), Q(LDQ,NMAX), W(NMAX),
+                WORK(7*NMAX), Z(LDZ,MMAX)
INTEGER          INDEX(NMAX), IWORK(5*NMAX)
*      .. External Subroutines ..
EXTERNAL         DSBEVX, X04CAF
*      .. Intrinsic Functions ..
INTRINSIC        MAX, MIN
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08HBF Example Program Results'
WRITE (NOUT,*)
Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, KD
IF (N.LE.NMAX .AND. KD.LE.KDMAX) THEN
*
*      Read the lower and upper bounds of the interval to be searched,
*      and read the upper or lower triangular part of the matrix A
*      from data file
*
READ (NIN,*) VL, VU
IF (UPLO.EQ.'U') THEN
READ (NIN,*) ((AB(KD+1+I-J,J),J=I,MIN(N,I+KD)),I=1,N)
ELSE IF (UPLO.EQ.'L') THEN
READ (NIN,*) ((AB(1+I-J,J),J=MAX(1,I-KD),I),I=1,N)
END IF
*
*      Set the absolute error tolerance for eigenvalues. With ABSTOL
*      set to zero, the default value is used instead
*
ABSTOL = ZERO
*
*      Solve the band symmetric eigenvalue problem
*
CALL DSBEVX('Vectors','Values in range',UPLO,N,KD,AB,LDAB,Q,
+          LDQ,VL,VU,IL,IU,ABSTOL,M,W,Z,LDZ,WORK,IWORK,INDEX,
+          INFO)
*
IF (INFO.GE.0 .AND. M.LE.MMAX) THEN
*
*      Print solution
*
WRITE (NOUT,99999) 'Number of eigenvalues found =', M
WRITE (NOUT,*)
WRITE (NOUT,*) 'Eigenvalues'
WRITE (NOUT,99998) (W(J),J=1,M)
*
IFAIL = 0
CALL X04CAF('General',' ',N,M,Z,LDZ,'Selected eigenvectors',

```

```

+           IFAIL)
      IF (INFO.GT.0) THEN
        WRITE (NOUT,99999)
+         'INFO eigenvectors failed to converge, INFO =', INFO
        WRITE (NOUT,*)
+         'Indices of eigenvectors that did not converge'
        WRITE (NOUT,99997) (INDEX(J),J=1,M)
      END IF
      ELSE IF (M.GT.MMAX) THEN
        WRITE (NOUT,99996) 'M greater than MMAX, M =', M,
+         ', MMAX =', MMAX
      ELSE
        WRITE (NOUT,99999) 'Failure in DSBEVX. INFO =', INFO
      END IF
    ELSE
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'NMAX and/or KDMAX too small'
    END IF
  STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (3X,(8F8.4))
99997 FORMAT (3X,(8I8))
99996 FORMAT (1X,A,I5,A,I5)
END

```

## 9.2 Program Data

F08HBF Example Program Data

```

5      2                      :Values of N and KD
-3.0   3.0                    :Values of VL and VU

1.0   2.0   3.0
      2.0   3.0   4.0
          3.0   4.0   5.0
              4.0   5.0
                  5.0 :End of matrix A

```

## 9.3 Program Results

F08HBF Example Program Results

Number of eigenvalues found = 2

Eigenvalues

-2.6633 1.7511

Selected eigenvectors

```

      1      2
1 -0.6238 -0.5635
2  0.2575  0.3896
3  0.5900 -0.4008
4 -0.4308  0.5581
5 -0.1039 -0.2421

```

---